


68HC11 LCD Lab

1. Connect the LCD to the 68HC11 EVBU Board as follows:



2. Create, assemble, make appropriate comments and save the original program shown below.

```
*      Program to use an LCD.
*      The LCD is connected to port C with the following bit assignments:
*      V+ :LCD2
*      Gnd :LCD1
*      Vref:LCD3
*      C0 :LCD11  data I/O to LCD data bit 4
*      C1 :LCD12  data I/O to LCD data bit 5
*      C2 :LCD13  data I/O to LCD data bit 6
*      C3 :LCD14  data I/O to LCD data bit 7
*      C4 :LCD4   RS (register select) L=control, H=data
*      C5 :LCD5   R/W H=read, L=write
*      C6 :LCD6   E (enable) H is enabled, transaction on H->L edge.
*      C7 -unused-
*
*      Port C is located at $1003
*      Data direction for Port C is address $1007.  L=input, H=output

*NOTE_ON    EQU    $90*
*NOTE_OFF   EQU    $80*

Init
    ORG    $0100
    LDX    #$1000
    BSET   $07,X    #$FF
    LDAB   #$08
    CLR    $04,X

LOOPB
    INC    $04,X
    LDAA   #$03
```

```

        JSR     WRITE_C_NIBBLE
        LDY     #$5000
LOOPY1  DEY
        BNE     LOOPY1
        DECB
        BNE     LOOPB
        JSR     Delay
        JSR     Delay
        CLR     $04,X
        LDY     #LCD_Init_Table
More_Init
        INC     $04,X
        LDAA    $00,Y
        BEQ     Init_Done
        JSR     Write_C_LCD
        INY
        BRA     More_Init
Init_Done
        JSR     Delay
        JSR     Delay
        CLR     $04,X
        LDY     #LCD_Test_Table
More_Test
        INC     $04,X
        LDAA    $00,Y
        BEQ     Test_Done
        JSR     Write_D_LCD
        JSR     Delay
        INY
        BRA     More_Test
Test_Done
        JSR     Delay
        JSR     Delay
        CLR     $04,X
Test_Done_2
        DEC     $04,X
        JSR     Delay
        BRA     Test_Done_2
LCD_Init_Table
        FCB     $28      ;Data length (bit 4) = 4-bit, Number of lines (bit 2) = 2
        FCB     $08      ;Display on (bit 2), Cursor on (bit 1), Blink Cursor (bit 0)
        FCB     $01      ;Clear Display
        FCB     $06      ;Entry Mode, I/D (bit 1) inc/dec, S (bit) Display Shift
        FCB     $0F      ;Display on (bit 2), Cursor on (bit 1), Blink Cursor (bit 0)
        FCB     $00
LCD_Test_Table
        FCC     'Hi ET Students'
        FCB     $00
Delay
        PSHY
        LDY     #$1000
D_LOOP  DEY
        BNE     D_LOOP
        PULY
        RTS
Write_C_LCD
        PSHA
        LSRA
        LSRA
        LSRA
        LSRA
        BSR     Wait4BF
        BSR     Write_C_NIBBLE
        PULA
        PSHA
        BSR     Write_C_NIBBLE
        PULA
        RTS
Write_D_LCD
        PSHA
        LSRA
        LSRA
        LSRA
        LSRA
        BSR     Wait4BF
        BSR     Write_D_NIBBLE
        PULA
        PSHA

```

```

        BSR      Write_D_NIBBLE
        PULA
        RTS
*      Write nibble is low 4 bits of A
Write_C_NIBBLE
        PSHA
        ANDA     #$0F      ;Set upper 4 bits to zero
        BSET     $07,X     #$FF      ;Set bits 0-3 of Port C to Output
        ORAA     #$40      ;E=H, R/W=L, RS=L
        STAA     $03,X
        ANDA     #$0F      ;E=L, R/W=L, RS=L
        STAA     $03,X
        BCLR     $07,X     #$0F      ;Set bits 0-3 of Port C to Input
        PULA
        RTS
Write_D_NIBBLE
        PSHA
        ANDA     #$0F      ;Set upper 4 bits to zero
        BSET     $07,X     #$FF      ;Set bits 0-3 of Port C to Output
        ORAA     #$50      ;E=H, R/W=L, RS=H
        STAA     $03,X
        ANDA     #$1F      ;E=L, R/W=L, RS=H
        STAA     $03,X
        BCLR     $07,X     #$0F      ;Set bits 0-3 of Port C to Input
        PULA
        RTS
Wait4BF
        JSR      Delay
        RTS
        JSR      Delay
        PSHA
        PSHB
        LDAA     #$F0
        STAA     $07,X     ;Set bits 0-3 of Port C to Input
LCD_Busy
        LDAB     #$60      ;E=H, R/W=H, RS=L
        STAB     $03,X
        LDAA     $03,X     ;Get high nibble, Busy flag bit 3 (7 in LCD)
        LDAB     #$40      ;E=L, R/W=H, RS=L
        STAB     $03,X
        LDAB     #$60      ;E=H, R/W=H, RS=L
        STAB     $03,X
        LDAB     #$40      ;E=L, R/W=H, RS=L
        STAB     $03,X
        ANDA     #$08      ;Look at BF, bit 3 in A
        BNE     LCD_Busy    ;If BF=H, then LCD is busy, loop...
        PULB
        PULA
        RTS

```

3. Execute the program and observe the operation. The LCD should now display "Hi ET Students".
4. Modify the program to display different a different string of characters, such as one's name.